#! /usr/bin/python

## @file
# This file is part of scons-tools.
#
# @author Sebastian Rettenberger <sebastian.rettenberger@tum.de>
#
# @copyright Copyright (c) 2016, Technische Universitaet Muenchen.
# All rights reserved.
#
# Redistribution and use in source and binary forms, with or without
# modification, are permitted provided that the following conditions are met:
#
# 1. Redistributions of source code must retain the above copyright notice,
#    this list of conditions and the following disclaimer.
#
# 2. Redistributions in binary form must reproduce the above copyright notice,
#    this list of conditions and the following disclaimer in the documentation
#    and/or other materials provided with the distribution.
#
# 3. Neither the name of the copyright holder nor the names of its
#    contributors may be used to endorse or promote products derived from this
#    software without specific prior written permission.
#
# THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS"
# AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE
# IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE
# ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE
# LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR
# CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF
# SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS
# INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN
# CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE)
# ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE
# POSSIBILITY OF SUCH DAMAGE.

import subprocess
import sys

import SCons.Action

import utils.checks

__version_build_template="""/*
* This file is automatically generated by the build process
* DO NOT EDIT!
*/

#ifndef VERSION_STRING
#define VERSION_STRING "%s";
#endif // VERSION_STRING
"""

__gitversion = None

def get(env):
	if env.GetOption('help') or env.GetOption('clean'):
		return

	global __gitversion
	if __gitversion:
		return __gitversion

	conf = env.Configure()
	utils.checks.addDefaultTests(conf)

	try:
		if not conf.CheckProg('git'):
			__gitversion = 'unknown'
			return

		kw = { 'stdin' : 'devnull',
			'stdout' : subprocess.PIPE,
			'stderr' : subprocess.PIPE,
			'universal_newlines' : True,
		}
		p = SCons.Action._subproc(env, ['git', 'describe', '--dirty= (modified)', '--always'], **kw)
		out,err = p.communicate()
		status = p.wait()
		if err:
			sys.stderr.write(unicode(err))

		if status:
			__gitversion = 'unknown'
		else:
			__gitversion = out.strip()
	finally:
		conf.Finish()
		return __gitversion

def generateHeader(env, target='version.h'):
	version = get(env)

	def createVersionFile(target, source, env):
		"""Generate the version file with the current version in it"""
		contents = __version_build_template % (version)
		fd = open(target[0].path, 'w')
		fd.write(contents)
		fd.close()
		return 0

	build_version = env.Command(target, [], SCons.Action.Action(createVersionFile))
	env.AlwaysBuild(build_version)

	return build_version
